

接線の方程式と逆関数

Copyright ©2007 by Shigeto R. Nishitani

具体例：

接線の方程式

関数 $f(x)=x^2-3x+5$ の $x=5$ での接線を求めよ。また元の関数と接線をプロットせよ。

まず、関数を定義

```
> f:=x->x^2-3*x+5;
f:=x→x2-3x+5 (2.1.1)
```

x_0 に接点の値をいれておく。

```
> x0:=5;
この微分をとる。
> diff(f(x),x);
2x-3 (2.1.2)
```

接線の傾き a を求める。

```
> a:=subs(x=x0,diff(f(x),x));
a:=7 (2.1.3)
```

傾き a で点 $(x_0,f(x_0))$ を通る直線の方程式を求め、それを式 $eq1$ とする。


```
> eq1:=y-f(x0)=a*(x-x0);
eq1:=y-15=7x-35 (2.1.4)
```

y について解き、それを式 $eq2$ とする。

```
> eq2:=solve(eq1,y);
eq2:=-20+7x (2.1.5)
```

2つの関数 $f(x)$, $eq2$ (これは x の関数とみなす)をプロット。

```
> plot([f(x),eq2],x);
```



逆関数

関数 $f(x)=x^2-3x+1$ の逆関数を求め、元の関数とともにプロットせよ。

まず、関数を定義

```
> restart;f:=x->x^2-3*x+1;
```

逆関数の定義 $x=f(y)$ として、式 $eq1$ とする。

```
> eq1:=x=f(y);
eq1:=x=y2-3y+1 (2.2.1)
```

y について解く。答えに2式が出ているのに注意。


```
> eq3:=solve(eq1,y);
eq3:=3/2+1/2*sqrt(5+4x), 3/2-1/2*sqrt(5+4x) (2.2.2)
```

答えのそれぞれは、 $eq3[1]$, $eq3[2]$ として参照できる。

```
> eq3[1];eq3[2];
3/2+1/2*sqrt(5+4x)
3/2-1/2*sqrt(5+4x) (2.2.3)
```

これらを適当にプロット。

```
> plot([f(x),eq3[1],eq3[2]],x=-5..10,y=-5..10);
```



演習

- 1 関数 $f(x)=-x^2-3x$ の $x=2$ での接線を求め、元の関数と接線をプロットせよ。
- 2 関数 $f(x)=\log(x^2+3)$ の $x=2$ での接線を求め、元の関数と接線をプロットせよ。
- 3 $\log(x)$ の逆関数を求め、元の関数とともにプロットせよ。
- 4 $\sin(x)$ の逆関数を求め、元の関数とともにプロットせよ。